42 Aero Camino, Goleta, CA 93117
Tel (805) 685-0066 | Fax (805) 685-0067
info@biopac.com | **www.biopac.com**

7/26/23

Application Note 296: *Continuous Monitoring of a Monotonic Signal*
Application to GASSYS3 Systems

This application note presents a way to monitor the convergence of gas concentration values in a GASSYS3. The response curve is simulated using an RC filter with a long time constant.

1.0 Motivation

The flow of calibration gases into the GASSYS3 chamber can be heuristically monitored where the operator can “eyeball” the signal until it appears to reach an asymptotically stable value. This note offers a technique through BBS where an algorithm determines when the asymptotically stable value is reached.

2.0 Hardware Setup

To mimic the behavior of the GASSYS3 asymptotically reaching a stable concentration level which resembles an exponential curve, an RC filter is fabricated to interface with the MP150 and UIM100C. The time constant is designed to be

$$t = R * C$$

where

R = 5.5 MΩ
C = 10 µF
t = 55 seconds

$$V_o = V_e^{-1/RC}$$



Figure 1a Schematic of RC high pass filter to mimic GASSYS3 concentration level response.



Figure 1b RC filter. Implementation of RC filter. Input → OUT0, V_{RC} → CH1, V_R → CH2

A step-function type signal is imparted to the input. A voltage will appear across the resistor where the signal will experience a monotonic decay in amplitude. This decay mimics the GASSYS3 gas concentration level response.

3.0 Sampling of the Response

The core algorithm used to sample and determine convergence to the asymptotic stable value is as follows (code is listed within the Appendix):

```
                    |
                    V
            OnStartAcquisition
        Stiimulator 0 On, set to 5 V
    Set Beginning time variable, T = ticks / 60
        Delay = 10 seconds
            End
                    |
                    V
            OnIdleAcquisition
Ending time variable, K = ticks / 60
d = K – T
if d > 10 ;10 second swath of data
        Edit SelectAll
        Set Hcursor Hcursor2-D, Hcursor2
        R = Abs(Slope)
        If R < 0.01
            MP100 StopAcq
        Endif
        T = ticks / 60  ;new beginning time
Endif  ;cycle through again
            End
                    |
                    V
            OnEndAcquisition
        MP100 Set Out 0, 0
      MP100 Set Stim 0, Off
        Call “Process”
            End
                    |
                    V
            Process
        Select Wave 2
        Edit SelectAll
    Set Hcursor HCursor2-1, HCursor2
Check mean value against passing metrics
            End
```

4.0 Sample Output

The following are sample graphical and journal outputs



Figure 2a 10-second sampling **during** acquisition



Figure 2b One-second sampling **at end** of acquisition



Figure 2c Journal output **at end** of acquisition measurement

Appendix | BBS Code for Convergence Monitoring

```
Lessons
OnOpenFile
;;Program that auto-convergences during an acquisition  V3

CloseAllGraphs

W$ = "Test"

New W$

Journal New

Select Window W$

Call "SetupChannels"

Call "Buttons"

End
Buttons
RemoveAllButtons
CreateButton "Start Acq" "StartAcq"
CreateButton "Exit" "Exit"

End
StartAcq
MP100 StartAcq

End
OnStartAcquisition
MP100 Set Stim 0, Off
MP100 Set Stim 0, On
MP100 Set Out0 5

T = ticks / 60
D = 10

End
OnIdleAcquisition
```

```
K = ticks / 60

d = K - T

if d > D

  Select Wave 2
  Edit SelectAll
  Set HCursor HCursor2 - D, HCursor2
   R = Abs(Slope)
   if R < 0.001
     MP100 StopAcq
   endif
   T = ticks / 60
endif

End
OnEndAcquisition
MP100 Set Out0 0
MP100 Set Stim 0, Off
Call "Process"
End
Exit
MP150 StopAcq
Call "EndButtons"
End
EndButtons
RemoveAllButtons
CreateButton "Close" "Close"
CreateButton "Again" "OnOpenFile"
End
Close
Select Window W$
Journal Close
CloseGraph
End
SetupChannels
Select Window W$

MP100 Set Channel AllOff
MP100 Set Channel 1, Analog, Enable, Plot, Off, "Source"
MP100 Set Scaling Analog 1, 1, 1, -1, -1, "V"
```

```
MP100 Set Channel 2, Analog, Enable, Plot, Off, "Filter Output"
MP100 Set Scaling Analog 2, 1, 1, -1, -1, "V"

MP100 Set Storage Memory NoRepeat, Off, Save
MP100 Set Rate 2000
MP100 Set Length 300, Seconds ; ~ 300 secs
MP100 Set OverwritePrompt Off

Set MeasurementPrecision 6

Set HScale 30

End
Process
Select Wave 2
Edit SelectAll
Set HCursor HCursor2 - 1, HCursor2

X# = Mean
A# = -0.02
B# = 0.02

Journal Append " Mean = "
Journal Append STR$(X#, 4)

Call "CheckRange"
Call "EndButtons"

End
CheckRange
;**************************
; CheckRange
;**************************
y = 0
if X# < A#
        y = 1
endif
if y = 0
        if X# > B#
                y = 1
        endif
```

```
endif
if y = 1
        Y = 0
        Journal Append "\t***** FAILED! ***** ( Need "
        Journal Append A#
        Journal Append ".."
        Journal Append B#
        Journal Append " to pass.)"
else
        Journal Append "\t      ...PASSED ("
        Journal Append A#
        Journal Append ".."
        Journal Append B#
        Journal Append ")"
endif
Journal Append "\r"
;
End
```